
The Skogsrå was a mythical creature of the forest that appears in the form
of a small, beautiful woman with a seemingly friendly temperament. However,
those who are enticed into following her into the forest are never seen
again.

This library attempts to improve the use of OS environment variables for
application configuration:
	Variable defaults.
	Automatic type casting of values.
	Automatic docs and spec generation.
	OS environment template generation.
	Posibility to override Elixir configuration.
	Runtime reloading.
	Setting variable's values at runtime.
	Fast cached values access by using :persistent_term as temporal storage.
	Custom variable bindings.
	YAML and JSON configuration providers for Elixir releases.

Small example
You would create a config module e.g:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "My hostname"
  app_env :my_hostname, :myapp, :hostname,
    default: "localhost"
end
Calling MyApp.Config.my_hostname() will retrieve the value for the
hostname in the following order:
	From the OS environment variable $MYAPP_HOSTNAME (can be overriden by the
option os_env).
	From the configuration file e.g:   config :myapp,
     hostname: "my.custom.host"

	From the default value if it exists (In this case, it would return
"localhost").

Available options
There are several options for configuring an environment variables:
	Option	Type	Default	Description
	default	any	nil	Sets the Default value for the variable.
	type	Skogsra.Env.type()	:any	Sets the explicit type for the variable.
	os_env	binary	autogenerated	Overrides automatically generated OS environment variable name.
	binding_order	Skogra.Env.bindings()	[:system, :config]	Sets the load order for variable binding.
	binding_skip	Skogra.Env.bindings()	[]	Which variable bindings should be skipped.
	required	boolean	false	Whether the variable is required or not.
	cached	boolean	true	Whether the variable should be cached or not.

IMPORTANT: Options skip_system: true and skip_config: true have been
deprecated in favour of binding_skip: [:system] and binding_skip: [:config]
respectively.

Additional topics:
	Types
	Automatic type casting.
	Explicit type casting.
	Custom types.
	Explicit OS environment variable names.
	Required variables.
	Overriding Elixir Configuration.
	Caching variables.
	Handling different environments.
	Setting and reloading variables.
	Automatic docs generation.
	Automatic spec generation.
	Automatic template generation.
	Custom variable bindings.
	Elixir release YAML and JSON config providers.
	Using with Hab.
	Installation.

Types
Skogsrå implements several built-in types. Some of them can be
inferred from the default value while others must be
set explicitly:
	Type	Can be inferred?	Description
	:any	Yes	Any Elixir term.
	:binary	Yes	A string e.g. "localhost".
	:integer	Yes	An integer e.g. 42.
	:neg_integer	No	A negative integer.
	:non_neg_integer	No	A non negative integer.
	:pos_integer	No	A positive integer.
	:float	Yes	A float e.g. 42.0.
	:boolean	Yes	A boolean e.g. false.
	:atom	Yes	An atom e.g. :ok.
	:module	No	An existent module e.g. Enum.
	:unsafe_module	No	A module e.g. SomeModule.
	Skogsra.Type implementation	No	See Custom types section.

Automatic type casting
If the default value is set (and no explicit type is found), the variable
value will be casted as the same type of the default value. For this to work,
the default value should be of the following types:
	:any
	:binary
	:integer
	:float
	:boolean
	:atom

e.g. in the following example, the value will be casted to :atom
automatically:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "My environment"
  app_env :my_environment, :myapp, :environment,
    default: :prod
end
If either of the system OS or the application environment variables are defined,
Skogsrå will try to cast their values to the default value's type which it
atom e.g:
iex(1)> System.get_env("MYAPP_ENVIRONMENT")
"staging"
iex(2)> MyApp.Config.my_environment()
{:ok, :staging}
or
iex(1)> Application.get_env(:myapp, :environment)
"staging"
iex(2)> MyApp.Config.my_environment()
{:ok, :staging}
Note: If the variable is already of the desired type, it won't be casted.

Explicit type casting
A type can be explicitly set. The available types are the following:
	:any (default).
	:binary.
	:integer.
	:neg_integer.
	:non_neg_integer.
	:pos_integer.
	:float.
	:boolean.
	:atom.
	:module (modules loaded in the system).
	:unsafe_module (modules that might or might not be loaded in the system)
	A module name with an implementation for the behaviour Skogsra.Type.

e.g. in the following example, the value will be casted to :pos_integer
according to the variable definition:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "Port"
  app_env :my_port, :myapp, :port,
    default: 4000,
    type: :pos_integer
end
If either of the system OS or the application environment variables are defined,
Skogsrå will try to cast their values to a positive integer.
iex(1)> System.get_env("MYAPP_PORT")
"42"
iex(2)> MyApp.Config.my_port()
{:ok, 42}
or
iex(1)> Application.get_env(:myapp, :port)
42
iex(2)> MyApp.Config.my_port()
{:ok, 42}
Custom types
Skogsrå's types can be extented by implementing Skogra.Type behaviour e.g. a
possible implementation for casting "1, 2, 3, 4" to [integer()] would be:
defmodule MyList do
  use Skogsra.Type

  @impl Skogsra.Type
  def cast(value)

  def cast(value) when is_binary(value) do
    list =
      value
      |> String.split(~r/,/)
      |> Stream.map(&String.trim/1)
      |> Enum.map(&String.to_integer/1)

    {:ok, list}
  end

  def cast(value) when is_list(value) do
    if Enum.all?(value, &is_integer/1), do: {:ok, value}, else: :error
  end

  def cast(_) do
    :error
  end
end
If then we define the following enviroment variable with Skogsrå:
defmodule MyApp.Config do
  use Skogsra

  app_env :my_integers, :myapp, :integers,
    type: MyList
end
If either of the system OS or the application environment variables are defined,
Skogsrå will try to cast their values using our implementation e.g:
iex(1)> System.get_env("MYAPP_INTEGERS")
"1, 2, 3"
iex(2)> MyApp.Config.my_integers()
{:ok, [1, 2, 3]}
or
iex(1)> Application.get_env(:myapp, :integers)
[1, 2, 3]
iex(2)> MyApp.Config.my_integers()
{:ok, [1, 2, 3]}
Important: The default value is not cast according to type.

Explicit OS environment variable names
Though Skogsrå automatically generates the names for the OS environment
variables, they can be overriden by using the option os_env e.g:
defmodule MyApp.Config do
  use Skogsra

  app_env :db_hostname, :myapp, [:postgres, :hostname],
    os_env: "PGHOST"
end
This will override the value $MYAPP_POSTGRES_HOSTNAME with $PGHOST e.g:
iex(1)> System.get_env("MYAPP_POSTGRES_HOSTNAME")
"unreachable"
iex(2)> System.get_env("PGHOST")
"reachable"
iex(3)> MyApp.Config.db_hostname()
{:ok, "reachable"}
Required variables
It is possible to set a environment variable as required with the required
option e.g:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "My port"
  app_env :my_port, :myapp, :port,
    required: true
end
If none of the system OS or the application environment variables are defined,
Skogsrå will return an error e.g:
iex(1)> System.get_env("MYAPP_PORT")
nil
iex(2)> Application.get_env(:myapp, :port)
nil
iex(3)> MyApp.Config.my_port()
{:error, "Variable port in app myapp is undefined"}
The module will also provide validate and validate! functions that can be
used in your application startup phase to verify that all required variables
are present e.g:
iex(1)> System.get_env("MYAPP_PORT")
nil
iex(2)> Application.get_env(:myapp, :port)
nil
iex(3)> MyApp.Config.validate!()
** (RuntimeError) Variable port in app myapp is undefined
Handling different environments
If it's necessary to keep several environments, it's possible to use a
namespace e.g. given the following variable:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "My port"
  app_env :my_port, :myapp, :port,
    default: 4000
end
Calling MyApp.Config.my_port(Test) will retrieve the value for the hostname
in the following order:
	From the OS environment variable $TEST_MYAPP_PORT.
	From the configuration file e.g:   config :myapp, Test,
     port: 4001

	From the OS environment variable $MYAPP_PORT.
	From the configuraton file e.g:   config :myapp,
     port: 80

	From the default value if it exists. In our example, 4000.

The ability of loading different environments allows us to do the following
with our configuration file:
# file: config/config.exs
use Mix.Config

config :myapp, Prod
  port: 80

config :myapp, Test,
  port: 4001

config :myapp,
  port: 4000
While our Skogsrå module would look like:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "Application environment"
  app_env :env, :myapp, :environment,
    type: :unsafe_module

  @envdoc "Application port"
  app_env :port, :myapp, :port,
    default: 4000
end
The we can retrieve the values depending on the value of the OS environment
variable $MYAPP_ENVIRONMENT:
...
with {:ok, env} <- MyApp.Config.env(),
     {:ok, port} <- Myapp.Config.port(env) do
  ... do something with the port ...
end
...
Overriding Elixir Configuration
Not all the libraries will use Skogsrå and, though is a shame, we can always
override a config value using the function preload/0 e.g.
Let's say we have a Phoenix application and we want to configure our endpoint
with the following variables.
	The OS environment variable PORT for our app's HTTP port.
	The OS environment variable SECRET_KEY_BASE for our secret key base.

A regular Phoenix configuration will look like the following:
use Mix.Config

secret_key_base =
  System.get_env("SECRET_KEY_BASE") ||
    raise """
    environment variable SECRET_KEY_BASE is missing.
    You can generate one by calling: mix phx.gen.secret
    """


config :myapp_web, MyappWeb.Endpoint,
  http: [
    port: String.to_integer(System.get_env("PORT") || "4000"),
    transport_options: [socket_opts: [:inet6]]
  ],
  secret_key_base: secret_key_base
With Skogsrå, we can reduce the config to the following:
use Mix.Config

config :myapp_web, Myapp.Endpoint,
  http: [
    transport_options: [socket_opts: [:inet6]]
  ]
Then, handle our variables in our Skogsrå config module:
defmodule MyappWeb.Config do
  use Skogsra

  @envdoc """
  App port.
  """
  app_env :port, :myapp, [:http, :port],
    os_env: "PORT",
    default: 4000

  @envdoc """
  Secret key base.
  """
  app_env :secret_key_base, :myapp, :secret_key_base,
    os_env: "SECRET_KEY_BASE",
    default: "+/JwMGGtsTVOoX5gQrCMn8aHKfKDdUK8GeAKJ2fIUabUnmWTwg+zsCy4pAOmOdTs"
end
And finally calling the preload function in our application init:
defmodule MyappWeb.Application do
  use Application

  def start(_type, _args) do
    MyappWeb.Config.preload(MyappWeb.Endpoint)

    children = [
      MyappWeb.Endpoint
    ]

    ...
  end
end
The function MyappWeb.Config.preload/1 will override any configuration for
MyappWeb.Endpoint as long as is a runtime config.
Our app now will try to get the port and secret key from the OS environment
variables first and it will fallback to the defaults.
Caching variables
By default, Skogsrå caches the values of the variables using
:persistent_term Erlang module. This makes reads very fast, but writes are
very slow.
So avoid setting or reloading values to avoid performance issues (see
Setting and reloading variables).
If you don't want to cache the values, you can set it to false:
defmodule MyApp.Config do
  use Skogsra

  app_env :value, :myapp, :value,
    cached: false
end
Setting and reloading variables
Every variable definition will generate two additional functions for setting
and reloading the values e.g:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "My port"
  app_env :my_port, :myapp, :port,
    default: 4000
end
will have the functions:
	MyApp.Config.put_my_port/1 for setting the value of the variable at
runtime.
	MyApp.Config.reload_my_port/ for reloading the value of the variable at
runtime.

Automatic docs generation
It's possible to document a configuration variables using the module attribute
@envdoc e.g:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "My port"
  app_env :my_port, :myapp, :port,
    default: 4000
end
Skogsra then will automatically generate instructions on how to use the
variable. This extra documentation can be disable with the following:
config :skogsra,
  generate_docs: false
Note: for "private" configuration variables you can use @envdoc false.

Automatic spec generation
Skogsra will try to generate the appropriate spec for every function generated.
In our example, given the default value is an integer, the function spec will be
the following:
@spec my_port() :: {:ok, integer()} | {:error, binary()}
@spec my_port(Skogsra.Env.namespace()) ::
        {:ok, integer()} | {:error, binary()}
Note: The same applies for my_port!/0, reload_my_port/0 and
put_my_port/1.

Specs are generated following the following table:
Assuming the type is set to :integer explicitly (type: :integer) or
implicitly (default: 4000), the specs should follow the following table:
	Required	Has default	Spec
	true	true	@spec my_port() :: {:ok, integer()} | {:error, binary()}
	true	false	@spec my_port() :: {:ok, integer()} | {:error, binary()}
	false	true	@spec my_port() :: {:ok, integer()} | {:error, binary()}
	false	false	@spec my_port() :: {:ok, nil | integer()} | {:error, binary()}

Automatic template generation
Every Skogsra module includes the functions template/1 and template/2 for
generating OS environment variable files e.g. continuing our example:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "My port"
  app_env :my_port, :myapp, :port,
    default: 4000
end
	For Elixir releases:
iex(1)> Myapp.Config.template("env")
:ok
will generate the file env with the following contents:
# DOCS My port
# TYPE integer
MYAPP_PORT="Elixir.Application"


	For Unix:
iex(1)> Myapp.Config.template("env", type: :unix)
:ok
will generate the file env with the following contents:
# DOCS My port
# TYPE integer
export MYAPP_PORT='Elixir.Application'


	For Windows:
iex(1)> Myapp.Config.template("env.bat", type: :windows)
:ok
will generate the file env.bat with the following contents:
:: DOCS My port
:: TYPE integer
SET MYAPP_PORT="Elixir.Application"


Custom variable bindings
By default, Skogsrå loads variables in the following order:
	OS environment variables or :system.
	Application configuratuon or :config.

This means that every variable has the following defaults:
	For binding_order is [:system, :config].
	For binding_skip is [].

These two values can be modified globally via configuration e.g:
	For globally changing the variable binding order:
 config :skogsra,
   binding_order: [:config, :system]

	For globally skipping :system:
 config :skogsra,
   binding_skip: [:system]


Or be modified per app_env e.g:
	For changing MyApp.Config.my_port/0 binding order:
 defmodule MyApp.Config do
   use Skogsra

   @envdoc "My port"
   app_env :my_port, :myapp, :port,
     binding_order: [:config, :system]
 end

	For skipping :system in MyApp.Config.my_port/0:
 defmodule MyApp.Config do
   use Skogsra

   @envdoc "My port"
   app_env :my_port, :myapp, :port,
     binding_skip: [:system]
 end


Additionally, we can create new variable bindings by implementing
Skogsra.Binding behaviour e.g. an implementation for loading JSON
configuration files would be:
defmodule MyApp.Json do
  use Skogsra.Binding

  alias Skogsra.Env

  @impl true
  def init(%Env{} = env) do
    options = Env.extra_options(env)

    case options[:config_path] do
      nil ->
        {:error, "JSON config path not specified"}

      path ->
        load(path)
    end
  end

  @impl true
  def get_env(%Env{} = env, config) when is_map(config) do
    name = Env.os_env(env)
    value = config[name]

    {:ok, value}
  end

  # Helpers

  # Loads JSON once and caches it in a :persistent_term using the path
  # as the key.
  @spec load(binary()) :: {:ok, map()} | {:error, term()}
  defp load(path) do
    with nil <- :persistent_term.get(path, nil),
         {:ok, contents} <- File.read(path),
         {:ok, config} <- Jason.decode(contents),
         :ok <- :persistent_term.put(path, config) do
      {:ok, config}
    else
      {:error, reason} ->
        {:error, "Cannot load #{path} due to #{inspect(reason)}"}

      config ->
        {:ok, config}
    end
  end
end
The previous implementation expects variables to be named the same as the
OS environment variables e.g:
{
  "MYAPP_PORT": 5000
}
Then our variable declaration would look something like the following:
defmodule MyApp.Config do
  use Skogsra

  @envdoc "My port"
  app_env :my_port, :myapp, :port,
    binding_order: [:system, :config, MyApp.Json],
    config_path: "#{Path.cwd!()}/priv/config.json",
    default: 4000
end
If no :system or :config is found, it will try to load the variable from
the JSON file e.g:
iex> MyApp.Config.port()
{:ok, 5000}
Note: The same casting rules apply for all bindings.

Elixir release YAML and JSON config providers
Skogsrå includes two simple configuration providers compatible with
mix release for Elixir ≥ 1.9:
	YAML configuration provider:
 - app: "my_app"              # Name of the application.
   module: "MyApp.Repo"       # Optional module/namespace.
   config:                    # Actual configuration.
     database: "my_app_db"
     username: "postgres"
     password: "postgres"
     hostname: "localhost"
     port: 5432

	JSON configuration provider:
 [
   {
     "app": "my_app",
     "module": "MyApp.Repo",
     "config": {
       "database": "my_app_db",
       "username": "postgres",
       "password": "postgres",
       "hostname": "localhost",
       "port": 5432
     }
   }
 ]


Both configurations would be equivalent to:
config :my_app, MyApp.Repo,
  database: "my_App_db",
  username: "postgres"
  password: "postgres"
  hostname: "localhost"
  port: 5432
For using these config providers, just add the following to your release
configuration:
	For YAML configurations:
 config_providers: [{Skogsra.Provider.Yaml, ["/path/to/config/file.yml"]}]

	For JSON configurations:
 config_providers: [{Skogsra.Provider.Json, ["/path/to/config/file.json"]}]


Note: If the module you're using in you're config does not exist, then
change it to namespace e.g: namespace: "MyApp.Repo". Otherwise, it will
fail loading it.

Using with Hab
Hab is an
Oh My ZSH plugin for loading OS
environment variables automatically.
By default, Hab will try to load .envrc file, but it's possible to have
several of those files for different purposes e.g:
	.envrc.prod for production OS variables.
	.envrc.test for testing OS variables.
	.envrc for development variables.

Hab will load the development variables by default, but it can load the
other files using the command hab_load <extension> e.g. loading
.envrc.prod would be as follows:
~/my_project $ hab_load prod
[SUCCESS]  Loaded hab [/home/user/my_project/.envrc.prod]

Installation
The package can be installed by adding skogsra to your list of dependencies
in mix.exs (Elixir ≥ 1.9 and Erlang ≥ 22):
def deps do
  [
    {:skogsra, "~> 2.4"}
  ]
end
If you need YAML config provider support, add the following:
def deps do
  [
    {:skogsra, "~> 2.4"},
    {:yamerl, "~> 0.10"}
  ]
end
If you need JSON config provider support, add  the following:
def deps do
  [
    {:skogsra, "~> 2.4"},
    {:jason, "~> 1.3"}
  ]
end
Author
Alexander de Sousa.
License
Skogsrå is released under the MIT License. See the LICENSE file for further
details.

Changelog for 2.4.0
Bugfix
	Bugfix: [YAML Provider] fixed bug where it would duplicate apps.
	Bugfix: [YAML Provider] fixed bug where regular lists weren't accepted as values.
	Migrated pipeline from Travis to Github Actions.
	Improved code coverage.
	Updated dependencies.

Changelog for 2.3.3
Enhancements
	Added :neg_integer, :non_neg_integer and :pos_integer as custom
types.

Changelog for 2.3.2
Bugfix
	Wrong spec in private function.

Changelog for 2.3.1
Bugfix
	Multiline docs in templates weren't properly commented.

Changelog for 2.3.0
Enhancements
	Added preload/0 and preload/1 functions for overriding Elixir config
values when config providers and OS system variables have precedence.

Changelog for 2.2.3
Bugfix
	When there is no default and it's not required, a variable should have the
variable's type and nil as return value e.g. for the variable :my_var
that returns integer, the spec would be as follows:
@spec my_var() :: {:ok, nil | integer()} | {:error, binary()}


Changelog for 2.2.2
Enhancements
	Added validate!/0 and validate!/1 functions for checking whether the
required variables have values or not (thanks to
Riccardo Binetti).
	Travis-CI build now checking format, credo, dialyzer and coveralls.

Changelog for 2.2.1
Bugfix
	When no type is defined and it cannot be derived from the default value,
then it should return any() instead of binary().

Changelog for 2.2.0
Enhancements
	Added JSON config provider.
	Added new Skogsra.Binding behaviour for adding custom variable bindings.
	New option binding_order for setting a custom order for variable binding
(defaults to [:system, :config]). Possible values:	:system: For loading OS environment variables.
	:config: For loading application configuration variables.
	module(): For loading variables using a custom module implementing
Skogsra.Binding behaviour.


	Added binding_skip option to skip one or several variable binding types
(defaults to []).

Breaking changes
	Improved YAML config provider by making it equivalent to the JSON provider.

Deprecations
	The option skip_config was deprecated favoring binding_skip: [:config].
	The option skip_system was deprecated favoring binding_skip: [:system].

Changelog for 2.1.1
Bug fixes
	Fixed bug where Skogsra.Type couldn't be used as a type (see Custom Type Fails at get_spec_type)

Changelog for 2.1.0
Enhancements
	Improved function specs.
	Improved generated docs.
	Added option to avoid automatically generated docs.
	Added function for OS environment variable generation for Unix, Releases
and Windows.

Changelog for 2.0.4
Enhancements
	Improved documentation.

Changelog for 2.0.3
Enhancements
	Improved errors when a variable is missing.

Changelog for 2.0.2
Enhancements
	Added :module built-in type.
	Added :unsafe_module built-in type.

Changelog for 2.0.0
Enhancements
	Added Skogsra.Type behaviour for defining custom types.
	Application configuration values are now casted as well to the defined
type in the Skogsra environment variable definition.
	Added YAML configuration provider.

Changelog for 1.3.0
Enhancements
	Namespaces now inherit values from default namespace.
	Favoring the use of :persistent_terms over :ets tables. This avoids the
creation of an application tree for this project, thus making it even
faster.

Changelog for 1.2.1
Bug fixes
	Fixed bug where Skogsra.Cache didn't work at compilation time.

Changelog for 1.2.0
Enhancements
	Added autogenerated reload functions.
	Added autogenerated setter functions.
	Improved documentation.
	Code rearrangement and refactor for maintainability.

This module defines the macros needed to use Skogsra e.g:
defmodule MyApp.Settings do
  use Skogsra

  @envdoc "My hostname"
  app_env :my_hostname, :myapp, :hostname,
    default: "localhost"
end

          Imports app_env/3 and app_env/4. Additionally generates the function

Creates a function to retrieve specific environment/application variables
values.
The function created is named function_name and will get the value
associated with an application called app_name and one or several
parameters keys. Optionally, receives a list of options.
Available options:
	Option	Type	Default	Description
	default	any	nil	Sets the Default value for the variable.
	type	Skogsra.Env.type()	:binary	Sets the explicit type for the variable.
	os_env	binary	autogenerated	Overrides automatically generated OS environment variable name.
	binding_order	Skogra.Env.bindings()	[:system, :config]	Sets the load order for variable binding.
	binding_skip	Skogra.Env.bindings()	[]	Which variable bindings should be skipped.
	required	boolean	false	Whether the variable is required or not.
	cached	boolean	true	Whether the variable should be cached or not.
	namespace	module	nil	Overrides any namespace.

e.g:
For the following declaration:
app_env :db_password, :myapp, [:mydb, :password],
  default: "password",
will generate:
	db_password/0 and db_password/1 for getting the variable's value
without or with namespace respectively. It returns :ok and :error
tuples.
	db_password!/0 and db_password!/1 for getting the variable's value
without or with namespace respectively. It fails on error.
	reload_db_password/0 and reload_db_password/1 for reloading the
variable's value in the cache without or with namespace respectively.
	put_db_password/1 and put_db_password/2 for settings a new value for
the variable directly to the cache without or with namespace respectively.

A call to db_password/0 will try to get a value:
	From the OS environment variable $MYAPP_MYDB_PASSWORD (can be overriden
by the option os_env).
	From the configuration file e.g:   config :myapp,
     mydb: [password: "some password"]

	From the default value if it exists (In this case, it would return
"password").

A call to db_password/1 with namespace Test will try to get a value:
	From the OS environment variable $TEST_MYAPP_MYDB_PASSWORD.
	From the configuration file e.g:   config :myapp, Test,
     mydb: [password: "some test password"]

	From the OS environment variable $MYAPP_MYDB_PASSWORDT.
	From the configuraton file e.g:   config :myapp,
     mydb: [password: "some password"]

	From the default value if it exists. In our example, "password".


  


        

      



  

    
Skogsra.Core 
    



      
This module defines the core API for Skogsra.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    
      Functions
    


      
        
          get_env!(env)

        


          Gets the value of a given env. Fails on error.



      


      
        
          get_env(env)

        


          Gets the value of a given env.



      


      
        
          put_env(env, value)

        


          Puts a new value for an env.



      


      
        
          reload_env(env)

        


          Reloads an env variable.



      


  


      


      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this function
    
    get_env!(env)


      
       
       View Source
     


  


  

      

          @spec get_env!(Skogsra.Env.t()) :: term() | no_return()


      


Gets the value of a given env. Fails on error.

  



  
    
      
      Link to this function
    
    get_env(env)


      
       
       View Source
     


  


  

      

          @spec get_env(Skogsra.Env.t()) :: {:ok, term()} | {:error, binary()}


      


Gets the value of a given env.

  



  
    
      
      Link to this function
    
    put_env(env, value)


      
       
       View Source
     


  


  

      

          @spec put_env(Skogsra.Env.t(), term()) :: :ok | {:error, binary()}


      


Puts a new value for an env.

  



  
    
      
      Link to this function
    
    reload_env(env)


      
       
       View Source
     


  


  

      

          @spec reload_env(Skogsra.Env.t()) :: {:ok, term()} | {:error, binary()}


      


Reloads an env variable.

  


        

      



  

    
Skogsra.Env 
    



      
This module defines a Skogsra environment variable.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    
      Types
    


      
        
          app_name()

        


          Application name.



      


      
        
          binding()

        


          Variable binding.



      


      
        
          bindings()

        


          Variable binding list.



      


      
        
          key()

        


          Key.



      


      
        
          keys()

        


          List of keys that lead to the value of the variable.



      


      
        
          namespace()

        


          Variable namespace.



      


      
        
          option()

        


          Environment variable options.



      


      
        
          options()

        


          Environment variable options



      


      
        
          t()

        


          Skogsra environment variable.



      


      
        
          type()

        


          Types.



      


  


  
    
      Functions
    


      
        
          %Skogsra.Env{}

        


          Environment variable struct.



      


      
        
          binding_order(env)

        


          Gets the binding order for a Skogsra environment variable.



      


      
        
          cached?(env)

        


          Whether the Skogsra environment variable is cached or not.



      


      
        
          default(env)

        


          Gets the default value for a Skogsra environment variable.



      


      
        
          extra_options(env)

        


          Gets extra options.



      


      
        
          new(namespace, app_name, keys, options)

        


          Creates a new Skogsra environment variable.



      


      
        
          os_env(env)

        


          Gets the OS variable name for the Skogsra environment variable.



      


      
        
          required?(env)

        


          Whether the Skogsra environment variable is required or not.



      


      
        
          type(env)

        


          Gets the type of the Skogsra environment variable.



      


  


      


      
        
          
            
            Anchor for this section
          
Types
        

        


  
    
      
      Link to this type
    
    app_name()


      
       
       View Source
     


  


  

      

          @type app_name() :: nil | atom()


      


Application name.

  



  
    
      
      Link to this type
    
    binding()


      
       
       View Source
     


  


  

      

          @type binding() :: :config | :system | module()


      


Variable binding.

  



  
    
      
      Link to this type
    
    bindings()


      
       
       View Source
     


  


  

      

          @type bindings() :: [binding()]


      


Variable binding list.

  



  
    
      
      Link to this type
    
    key()


      
       
       View Source
     


  


  

      

          @type key() :: atom()


      


Key.

  



  
    
      
      Link to this type
    
    keys()


      
       
       View Source
     


  


  

      

          @type keys() :: [key()]


      


List of keys that lead to the value of the variable.

  



  
    
      
      Link to this type
    
    namespace()


      
       
       View Source
     


  


  

      

          @type namespace() :: nil | atom()


      


Variable namespace.

  



  
    
      
      Link to this type
    
    option()


      
       
       View Source
     


  


  

      

          @type option() ::
  {:binding_order, bindings()}
  | {:binding_skip, bindings()}
  | {:os_env, binary()}
  | {:type, type()}
  | {:namespace, namespace()}
  | {:default, term()}
  | {:required, boolean()}
  | {:cached, boolean()}
  | {atom(), term()}


      


Environment variable options.
	binding_order - Variable binding load order.
	binding_skip - Skips loading a variable from the list of bindings.
	os_env - The name of the OS environment variable.
	type - Type to cast the OS environment variable value.
	namespace - Default namespace for the variable.
	default - Default value.
	required - Whether the variable is required or not.
	cached - Whether the variable is cached or not.


  



  
    
      
      Link to this type
    
    options()


      
       
       View Source
     


  


  

      

          @type options() :: [option()]


      


Environment variable options:

  



  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %Skogsra.Env{
  app_name: app_name :: app_name(),
  keys: keys :: keys(),
  namespace: namespace :: namespace(),
  options: options :: options()
}


      


Skogsra environment variable.

  



  
    
      
      Link to this type
    
    type()


      
       
       View Source
     


  


  

      

          @type type() ::
  :binary
  | :integer
  | :neg_integer
  | :non_neg_integer
  | :pos_integer
  | :float
  | :boolean
  | :atom
  | :module
  | :unsafe_module
  | :any
  | module()


      


Types.

      Link to this function
    
    %Skogsra.Env{}


      
       
       View Source
     


      (struct)

  


  

Environment variable struct.

  



  
    
      
      Link to this function
    
    binding_order(env)


      
       
       View Source
     


  


  

      

          @spec binding_order(t()) :: bindings()


      


Gets the binding order for a Skogsra environment variable.

  



  
    
      
      Link to this function
    
    cached?(env)


      
       
       View Source
     


  


  

      

          @spec cached?(t()) :: boolean()


      


Whether the Skogsra environment variable is cached or not.

  



  
    
      
      Link to this function
    
    default(env)


      
       
       View Source
     


  


  

      

          @spec default(t()) :: term()


      


Gets the default value for a Skogsra environment variable.

  



  
    
      
      Link to this function
    
    extra_options(env)


      
       
       View Source
     


  


  

      

          @spec extra_options(t()) :: keyword()


      


Gets extra options.

  



  
    
      
      Link to this function
    
    new(namespace, app_name, keys, options)


      
       
       View Source
     


  


  

      

          @spec new(namespace(), app_name(), key(), options()) :: t()


          @spec new(namespace(), app_name(), keys(), options()) :: t()


      


Creates a new Skogsra environment variable.

  



  
    
      
      Link to this function
    
    os_env(env)


      
       
       View Source
     


  


  

      

          @spec os_env(t()) :: binary()


      


Gets the OS variable name for the Skogsra environment variable.

  



  
    
      
      Link to this function
    
    required?(env)


      
       
       View Source
     


  


  

      

          @spec required?(t()) :: boolean()


      


Whether the Skogsra environment variable is required or not.

  



  
    
      
      Link to this function
    
    type(env)


      
       
       View Source
     


  


  

      

          @spec type(t()) :: type() | tuple()


      


Gets the type of the Skogsra environment variable.

  


        

      



  

    
Skogsra.Cache 
    



      
This module defines the helpers for the variable cache. The cache works with
persistent terms.

          Summary
        


  
    
      Functions
    


      
        
          get_env(env)

        


          Gets a variable value env from the cache.



      


      
        
          put_env(env, value)

        


          Stores a variable env in a persistent term.

Functions
        

        


  
    
      
      Link to this function
    
    get_env(env)


      
       
       View Source
     


  


  

      

          @spec get_env(Skogsra.Env.t()) :: {:ok, term()} | :error


      


Gets a variable value env from the cache.

  



  
    
      
      Link to this function
    
    put_env(env, value)


      
       
       View Source
     


  


  

      

          @spec put_env(Skogsra.Env.t(), term()) :: :ok


      


Stores a variable env in a persistent term.

  


        

      



  

    
Skogsra.Type behaviour
    



      
This module defines the functions and behaviours for casting Skogsra types.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    
      Callbacks
    


      
        
          cast(term)

        


          Callback for casting a value.



      


  


  
    
      Functions
    


      
        
          __using__(_)

        


          Uses Skogsra.Type for implementing the behaviour e.g. a naive implementation
for casting "1, 2, 3, 4" to ["1", "2", "3", "4"] would be



      


      
        
          cast(env, value)

        


          Casts an environment variable.



      


  


      


      
        
          
            
            Anchor for this section
          
Callbacks
        

        


  
    
      
      Link to this callback
    
    cast(term)


      
       
       View Source
     


  


  

      

          @callback cast(term()) :: {:ok, term()} | :error


      


Callback for casting a value.

  


        

      

      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this macro
    
    __using__(_)


      
       
       View Source
     


      (macro)

  


  

Uses Skogsra.Type for implementing the behaviour e.g. a naive implementation
for casting "1, 2, 3, 4" to ["1", "2", "3", "4"] would be:
defmodule MyList do
  use Skogsra.Type

  def cast(value) when is_binary(value) do
    list =
      value
      |> String.split(~r/,/)
      |> Enum.map(&String.trim/1)
    {:ok, list}
  end

  def cast(_) do
    :error
  end
end

  



  
    
      
      Link to this function
    
    cast(env, value)


      
       
       View Source
     


  


  

      

          @spec cast(Skogsra.Env.t(), term()) :: {:ok, term()} | :error


      


Casts an environment variable.

  


        

      



  

    
Skogsra.App 
    



      
This module defines the functions to get variables from the application
configuration.

          Summary
        


  
    
      Functions
    


      
        
          preload(env)

        


          Overrides an config value with Skogsra's value.



      


  


      


      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this function
    
    preload(env)


      
       
       View Source
     


  


  

      

          @spec preload(Skogsra.Env.t()) :: :ok


      


Overrides an config value with Skogsra's value.

  


        

      



  

    
Skogsra.Binding behaviour
    



      
Variable binding behaviour.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    
      Callbacks
    


      
        
          get_env(env, config)

        


          Callback for getting an environment variable.



      


      
        
          init(env)

        


          Callback for initializing binding.



      


  


  
    
      Functions
    


      
        
          __using__(_)

        


          Uses the Skogsra.Binding behaviour.



      


      
        
          get_env(module, env)

Callbacks
        

        


  
    
      
      Link to this callback
    
    get_env(env, config)


      
       
       View Source
     


  


  

      

          @callback get_env(env :: Skogsra.Env.t(), config :: term()) ::
  {:ok, term()} | {:error, term()}


      


Callback for getting an environment variable.

  



  
    
      
      Link to this callback
    
    init(env)


      
       
       View Source
     


  


  

      

          @callback init(env :: Skogsra.Env.t()) :: {:ok, term()} | {:error, term()}


      


Callback for initializing binding.

Functions
        

        


  
    
      
      Link to this macro
    
    __using__(_)


      
       
       View Source
     


      (macro)

  


  

Uses the Skogsra.Binding behaviour.

  



  
    
      
      Link to this function
    
    get_env(module, env)


      
       
       View Source
     


  


  

      

          @spec get_env(Skogsra.Env.binding(), env :: Skogsra.Env.t()) :: term()


      



  


        

      



  

    
Skogsra.Sys 
    



      
This module defines the functions to get environment variables from the OS.

      





  

    
Skogsra.Docs 
    



      
This module defines the documentation generators.

          Summary
        


  
    
      Types
    


      
        
          docs()

        


          Custom docs.



      


      
        
          function_name()

        


          Function name.



      


  


  
    
      Functions
    


      
        
          gen_full_docs(module, function_name, app_name, keys, options, docs)

        


          Generates docs for a variable given its module, function_name,
app_name, keys to find it, options and custom docs.



      


      
        
          gen_put_docs(module, function_name, docs)

        


          Generates put docs for a variable given its module and function_name.



      


      
        
          gen_reload_docs(module, function_name, docs)

        


          Generates reload docs for a variable given its module and function_name.



      


      
        
          gen_short_docs(module, function_name, docs)

        


          Generates short docs for a variable given its module, function_name,
and custom docs.



      


  


      


      
        
          
            
            Anchor for this section
          
Types
        

        


  
    
      
      Link to this type
    
    docs()


      
       
       View Source
     


  


  

      

          @type docs() :: nil | false | binary()


      


Custom docs.

  



  
    
      
      Link to this type
    
    function_name()


      
       
       View Source
     


  


  

      

          @type function_name() :: atom()


      


Function name.

Functions
        

        


  
    
      
      Link to this function
    
    gen_full_docs(module, function_name, app_name, keys, options, docs)


      
       
       View Source
     


  


  

      

          @spec gen_full_docs(
  module(),
  function_name(),
  Skogsra.Env.app_name(),
  Skogsra.Env.keys(),
  Skogsra.Env.options(),
  docs()
) :: binary()


      


Generates docs for a variable given its module, function_name,
app_name, keys to find it, options and custom docs.

  



  
    
      
      Link to this function
    
    gen_put_docs(module, function_name, docs)


      
       
       View Source
     


  


  

      

          @spec gen_put_docs(
  module(),
  function_name(),
  docs()
) :: binary()


      


Generates put docs for a variable given its module and function_name.

  



  
    
      
      Link to this function
    
    gen_reload_docs(module, function_name, docs)


      
       
       View Source
     


  


  

      

          @spec gen_reload_docs(
  module(),
  function_name(),
  docs()
) :: binary()


      


Generates reload docs for a variable given its module and function_name.

  



  
    
      
      Link to this function
    
    gen_short_docs(module, function_name, docs)


      
       
       View Source
     


  


  

      

          @spec gen_short_docs(
  module(),
  function_name(),
  docs()
) :: binary()


      


Generates short docs for a variable given its module, function_name,
and custom docs.

  


        

      



  

    
Skogsra.Spec 
    



      
This module defines the spec generators.

          Summary
        


  
    
      Functions
    


      
        
          gen_bang_spec(function_name, options)

        


          Generates spec for a bang function.



      


      
        
          gen_full_spec(function_name, options)

        


          Generates spec for a configuration function.



      


      
        
          gen_put_spec(function_name, options)

        


          Generates put spec for a configuration function.



      


      
        
          gen_reload_spec(function_name, options)

        


          Generates reload spec for a configuration function.

Functions
        

        


  
    
      
      Link to this function
    
    gen_bang_spec(function_name, options)


      
       
       View Source
     


  


  

Generates spec for a bang function.

  



  
    
      
      Link to this function
    
    gen_full_spec(function_name, options)


      
       
       View Source
     


  


  

Generates spec for a configuration function.

  



  
    
      
      Link to this function
    
    gen_put_spec(function_name, options)


      
       
       View Source
     


  


  

Generates put spec for a configuration function.

  



  
    
      
      Link to this function
    
    gen_reload_spec(function_name, options)


      
       
       View Source
     


  


  

Generates reload spec for a configuration function.

  


        

      



  

    
Skogsra.Template 
    



      
This module defines several templates for OS environment variable definitions.

          Summary
        


  
    
      Types
    


      
        
          t()

        


          A template struct.



      


      
        
          type()

        


          Template type.



      


  


  
    
      Functions
    


      
        
          %Skogsra.Template{}

        


          Template internal structure.



      


      
        
          generate(templates, filename)

        


          Generates a template.



      


      
        
          new(map)

        


          Builds a template struct from a map.

      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %Skogsra.Template{
  docs: docs :: Skogsra.Docs.docs(),
  env: env :: Skogsra.Env.t(),
  type: type :: type()
}


      


A template struct.

  



  
    
      
      Link to this type
    
    type()


      
       
       View Source
     


  


  

      

          @type type() :: :elixir | :unix | :windows


      


Template type.

      Link to this function
    
    %Skogsra.Template{}


      
       
       View Source
     


      (struct)

  


  

Template internal structure.

  



  
    
      
      Link to this function
    
    generate(templates, filename)


      
       
       View Source
     


  


  

      

          @spec generate([t()], Path.t()) :: :ok | {:error, File.posix()}


      


Generates a template.

  



  
    
      
      Link to this function
    
    new(map)


      
       
       View Source
     


  


  

      

          @spec new(map()) :: t()


      


Builds a template struct from a map.

  


        

      



  

    
Skogsra.Provider.Json 
    



      
This module defines a JSON config provider for Skogsra.
Important: You need to add {:jason, "~> 1.1"} as a dependency
along with Skogsra because the dependency is optional.

The following is an example for Ecto configuration in a file called
/etc/my_app/config.yml:
[
  {
    "app": "my_app",
    "module": "MyApp.Repo",
    "config": {
      "database": "my_app_db",
      "username": "postgres",
      "password": "postgres",
      "hostname": "localhost",
      "port": 5432
    }
  }
]
Then in your release configuration you can add the following:
config_providers: [{Skogsra.Provider.Json, ["/etc/my_app/config.json"]}]
Once the system boots, it'll parse and add the JSON configuration.

      





  

    
Skogsra.Provider.Yaml 
    



      
This module defines a YAML config provider for Skogsra.
Important: You need to add {:yamerl, "~> 0.7"} as a dependency
along with Skogsra because the dependency is optional.

The following is an example for Ecto configuration:
# file: /etc/my_app/config.yml
- app: "my_app"
  module: "MyApp.Repo"
  config:
    database: "my_app_db"
    username: "postgres"
    password: "postgres"
    hostname: "localhost"
    port: 5432
Then in your release configuration you can add the following:
config_providers: [{Skogsra.Provider.Yaml, ["/etc/my_app/config.yml"]}]
Once the system boots, it'll parse and add the YAML configuration.

      Types
    


      
        
          key()

      Link to this type
    
    key()


      
       
       View Source
     


  


  

      

          @type key() :: [integer()]

